
    Changelog

All notable changes to this project will be documented in this file.
The format is based on Keep a Changelog,
and this project adheres to Semantic Versioning.
1.0.2 (2026-06-06)
Bugfixes
	There was a bug where running with test coverage changed the fallback merge behavior for structs locally. The detection mechanism has been changed and the tests for those circumstances fixed.

1.0.1 (2026-06-04)
Bugfixes
	Fixed a wrong type spec, thanks martinmaillard

Documentation
	broad sweep improvement of docs, thanks kianmeng

1.0.0 (2019-03-26)
	Added the possibility to @derive [DeepMerge.Resolver] if you want your custom structs to also easily be deep_mergeable
	added type specs
	Changed elixir support to ~> 1.6

Considered feature complete (tm) - unless you come with great feature ideas of course ;)
0.2.0 (2018-07-27)
	Remove compile warnings on elixir 1.7
	Remove 1.2 from the build matrix (was getting big, should still work)

0.1.1 (2016-11-29)
	Remove warnings when run with elixir 1.4.0-rc.0

Provides functionality for "deep merging" maps and keyword lists in elixir, which is if during merging both values at the same key are maps/keyword lists merge them recursively. This is done via a protocol so can be extended for your own structs/data types if needbe.
iex> DeepMerge.deep_merge(%{a: 1, b: [x: 10, y: 9]}, %{b: [y: 20, z: 30], c: 4})
%{a: 1, b: [x: 10, y: 20, z: 30], c: 4}
This functionality can be useful for instance when merging a default configuration with a user supplied custom configuration:
DeepMerge.deep_merge(default_config, custom_config) # => merged configuration
Further features include:
	It handles both maps and keyword lists
	It does not merge structs or maps with structs…
	…but you can implement the simple DeepMerge.Resolver protocol for types/structs of your choice to also make them deep mergable (there is also a default implementation)
	a deep_merge/3 variant that gets a function similar to Map.merge/3 to modify the merging behavior, for instance in case you don't want keyword lists to be merged or you want all lists to be appended

I wanted this to be a feature of Elixir itself, however the proposal was rejected hence this library exists :)
Installation
Add :deep_merge to your list of dependencies in mix.exs:
def deps do
  [
    {:deep_merge, "~> 1.0"}
  ]
end
General Usage - deep_merge/2
Using this library is quite simple (and you might also want to look at the hexdocs) - just pass two structures to be deep merged into DeepMerge.deep_merge/2:
iex> DeepMerge.deep_merge(%{a: 1, b: %{x: 10, y: 9}}, %{b: %{y: 20, z: 30}, c: 4})
%{a: 1, b: %{x: 10, y: 20, z: 30}, c: 4}

iex> DeepMerge.deep_merge([a: 1, b: [x: 10, y: 9]], [b: [y: 20, z: 30], c: 4])
[a: 1, b: [x: 10, y: 20, z: 30], c: 4]
It is worth noting that structs are not deeply merged - not with each other and not with normal maps. This is because structs, while internally a map, are more like their own data types and therefore should not be deeply merged... unless you implement the protocol provided by this library for them.
Customization via protocols
What is merged and how is defined by implementing the DeepMerge.Resolver protocol. This library implements it for Map, List and falls back to Any (where the right hand side value/override is taken).
If you want your own struct to be deeply merged you can simply @derive the protocol:
defmodule Derived do
  @derive [DeepMerge.Resolver]
  defstruct [:attrs]
end
If you want to change the deep merge for a custom struct you can do so. An example implementation might look like this if you want to deeply merge your struct but only against non nil values (because all keys are always there) if you merge against the same struct (but still merge with maps):
defimpl DeepMerge.Resolver, for: MyStruct do
  def resolve(original, override = %MyStruct{}, resolver) do
    cleaned_override =
      override
      |> Map.from_struct()
      |> Enum.reject(fn {_key, value} -> is_nil(value) end)
      |> Map.new()

    Map.merge(original, cleaned_override, resolver)
  end

  def resolve(original, override, resolver) when is_map(override) do
    Map.merge(original, override, resolver)
  end
end
In this implementation, MyStruct structs are merged with other MyStruct structs, omitting nil values. The arguments passed to resolve are the original value (left hand side) and the override value (right hand side, which would normally replace the original). The third parameter is a resolver function which you can pass to Map.merge/3/Keyword.merge/3 to continue the deep merge.
Customization via deep_merge/3
There is another deep merge variant that is a bit like Map.merge/3 as it takes an additional function which you can use to alter the deep merge behavior:
iex> resolver = fn
...> (_, original, override) when is_list(original) and is_list(override) ->
...>   override
...> (_, _original, _override) ->
...>   DeepMerge.continue_deep_merge
...> end
iex> DeepMerge.deep_merge(%{a: %{b: 1}, c: [d: 1]},
...> %{a: %{z: 5}, c: [x: 0]}, resolver)
%{a: %{b: 1, z: 5}, c: [x: 0]}
This function is called for a given merge conflict with the key where it occurred and the two conflicting values. Whatever value is returned in this function is inserted at that point in the structure - unless DeepMerge.continue_deep_merge is returned in which case the deep merge continues as normal.
When would you want to use this versus a protocol? The best use case I can think of is when you want to alter behavior for which a protocol is already implemented or if you care about specific keys.
In the example above the behavior is changed so that keyword lists are not deep_merged (if they were the result would contain c: [d: 1, x:0]), but maps still are if that's what you are looking for.
Do I really need a library for this?
Well not necessarily, no. There are very simple implementations for maps that use Map.merge/3.
There are subtle things that can be missed there though (and I missed the first time around):
	the most simple implementation also merges structs which is not always what you want
	for keyword lists on the other hand you gotta be careful that you don't accidentally merge keyword lists with lists as that's currently possible
	you might want to further adopt the implementation, in benchee we have 2 custom implementations of the protocol due to our needs

This library takes care of those problems and will take care of further problems/edge cases should they appear so you can focus on your business logic.
At the same time it offers extension mechanisms through protocols and a function in deep_merge/3. So, it should be adjustable to your use case and if not please open an issue :)
Performance
You can check out a benchmark and its results.
The TLDR; is this: In the sample it is about 30 times slower than Map.merge/2 - however, less than twice as slow as calling Map.merge/3 with simple overriding behaviour (same behaviour as Map.merge/2). This is because Map.merge/2 is highly optimized, but we need to do much more than the Map.merge/3 sample in the benchmark so I think it's a very passable result. We're still talking about a couple of μs.
Considered feature-complete
Unless you come with great feature ideas of course ;) So if you come here and there are no recent commits, don't worry - there are no known bugs or whatever. It's a small little library that does its job.
Copyright and License
Copyright (c) 2016 Tobias Pfeiffer
This library is MIT licensed. See the
LICENSE for details.

Provides functionality for deeply/recursively merging structures (normally for
Map and Keyword).
If you want to change the deep merge behavior of a custom struct,
please have a look at the DeepMerge.Resolver protocol.

        continue_deep_merge()

      


        The symbol to return in the function in deep_merge/3 when deep merging
should continue as normal.



    


    
      
        deep_merge(original, override)

      


        Deeply merges two maps or keyword list original and override.



    


    
      
        deep_merge(original, override, resolve_function)

      


        A variant of DeepMerge.deep_merge/2 that allows to modify the merge behavior
through the additional passed in function.

          @spec continue_deep_merge() :: :__deep_merge_continue


      


The symbol to return in the function in deep_merge/3 when deep merging
should continue as normal.
Examples
iex> DeepMerge.continue_deep_merge
:__deep_merge_continue

          @spec deep_merge(map() | keyword(), map() | keyword()) :: map() | keyword()


      


Deeply merges two maps or keyword list original and override.
In more detail, if two conflicting values are maps or keyword lists themselves
then they will also be merged recursively. This is an extension in that sense
to what Map.merge/2 and Keyword.merge/2 do as it doesn't just override map
or keyword values but tries to merge them.
It does not merge structs or structs with maps. If you want your structs to be
merged then please have a look at the DeepMerge.Resolver protocol and
consider implementing/deriving it.
Also, while it says Map and Keyword here, it is really dependent on which
types implement the DeepMerge.Resolver protocol, which by default are Map
and Keyword.
Examples
iex> DeepMerge.deep_merge(%{a: 1, b: [x: 10, y: 9]}, %{b: [y: 20, z: 30], c: 4})
%{a: 1, b: [x: 10, y: 20, z: 30], c: 4}

iex> DeepMerge.deep_merge(%{a: 1, b: %{x: 10, y: 9}}, %{b: %{y: 20, z: 30}, c: 4})
%{a: 1, b: %{x: 10, y: 20, z: 30}, c: 4}

iex> DeepMerge.deep_merge(%{a: 1, b: %{x: 10, y: 9}}, %{b: %{y: 20, z: 30}, c: 4})
%{a: 1, b: %{x: 10, y: 20, z: 30}, c: 4}

iex> DeepMerge.deep_merge([a: 1, b: [x: 10, y: 9]], [b: [y: 20, z: 30], c: 4])
[a: 1, b: [x: 10, y: 20, z: 30], c: 4]

iex> DeepMerge.deep_merge(%{a: 1, b: 2}, %{b: 3, c: 4})
%{a: 1, b: 3, c: 4}

iex> DeepMerge.deep_merge(%{a: 1, b: %{x: 10, y: 9}}, %{b: 5, c: 4})
%{a: 1, b: 5, c: 4}

iex> DeepMerge.deep_merge([a: [b: [c: 1, d: 2], e: [24]]], [a: [b: [f: 3], e: [42, 100]]])
[a: [b: [c: 1, d: 2, f: 3], e: [42, 100]]]

iex> DeepMerge.deep_merge(%{a: 1, b: 5}, %{b: %{x: 10, y: 9}, c: 4})
%{a: 1, b: %{x: 10, y: 9}, c: 4}

iex> DeepMerge.deep_merge(%{a: [b: %{c: [d: "foo", e: 2]}]}, %{a: [b: %{c: [d: "bar"]}]})
%{a: [b: %{c: [e: 2, d: "bar"]}]}

  



  
    
      
    
    
      deep_merge(original, override, resolve_function)



        
          
        

    

  


  

      

          @spec deep_merge(map() | keyword(), map() | keyword(), (any(), any(), any() -> any())) ::
  map() | keyword()


      


A variant of DeepMerge.deep_merge/2 that allows to modify the merge behavior
through the additional passed in function.
This is similar to the relationship between Map.merge/2 and Map.merge/3
and the structure of the function is exactly the same, e.g. the passed in
arguments are key, original and override.
The function is called before a merge is performed. If it returns any value
that value is inserted at that point during the deep_merge. If the deep merge
should continue like normal you need to return the symbol returned by
DeepMerge.continue_deep_merge/0.
If the merge conflict occurs at the top level then key is nil.
The example shows how this can be used to modify deep_merge not to merge
keyword lists, in case you don't like that behavior.
Examples
iex> resolver = fn
...> (_, original, override) when is_list(original) and is_list(override) ->
...>   override
...> (_, _original, _override) ->
...>   DeepMerge.continue_deep_merge
...> end
iex> DeepMerge.deep_merge(%{a: %{b: 1}, c: [d: 1]},
...> %{a: %{z: 5}, c: [x: 0]}, resolver)
%{a: %{b: 1, z: 5}, c: [x: 0]}

  


        

      


  

    
DeepMerge.Resolver protocol
    



      
Protocol defining how conflicts during deep_merge should be resolved.
As part of the DeepMerge library this protocol is already implemented for
Map and List as well as a fallback to Any (which just always takes the
override).
If you want your custom structs to also be deeply mergable and not just
override one another (default behaviour) you can derive the protocol:
defmodule Derived do
  @derive [DeepMerge.Resolver]
  defstruct [:attrs]
end
It will then automatically be deeply merged with structs of its own kind, not
with other structs or maps though.

      


      
        Summary


  
    Types
  


    
      
        t()

      


        All the types that implement this protocol.



    





  
    Functions
  


    
      
        resolve(original, override, resolver)

      


        Defines what happens when a merge conflict occurs on this struct during a
deep_merge.

          @type t() :: term()


      


All the types that implement this protocol.

      resolve(original, override, resolver)



        
          
        

    

  


  

Defines what happens when a merge conflict occurs on this struct during a
deep_merge.
Can be implemented for additional data types to implement custom deep merging
behavior.
The passed in values are:
	original - the value in the original data structure, usually left side
argument
	override - the value with which original would be overridden in a
normal Map.merge/2
	resolver - the function used by DeepMerge to resolve merge conflicts,
i.e. what you can pass to Map.merge/3 and Keyword.merge/3 to continue
deeply merging.

An example implementation might look like this if you want to deeply merge
your struct but only against non nil values (because all keys are always there)
if you merge against the same struct (but still merge with maps):
defimpl DeepMerge.Resolver, for: MyStruct do
  def resolve(original, override = %MyStruct{}, resolver) do
    cleaned_override =
      override
      |> Map.from_struct()
      |> Enum.reject(fn {_key, value} -> is_nil(value) end)
      |> Map.new()

    Map.merge(original, cleaned_override, resolver)
  end

  def resolve(original, override, resolver) when is_map(override) do
    Map.merge(original, override, resolver)
  end
end

